TP9B : Les quaternions de Hamilton

Eléments de corrigé

```
> restart: with(LinearAlgebra):
```

Question 1

```
> Hadd:=(q1,q2)->[q1[1]+q2[1],VectorAdd(q1[2],q2[2])]:
> Hscal:=(lambda,q)->[lambda*q[1],VectorScalarMultiply(q[2],lam
  bda)]:
> Hmul:=(q1,q2)->[q1[1]*q2[1]-DotProduct(q1[2],q2[2],conjugate=
  false),VectorAdd(VectorAdd(q1[2],q2[2],q2[1],q1[1]),CrossProd
  uct(q1[2],q2[2]))]:
> e:=[1,<0,0,0>]: i:=[0,<1,0,0>]: j:=[0,<0,1,0>]:
  k:=[0,<0,0,1>]:
> Hadd(e,i);
```

$$\left[1, \begin{bmatrix}1\\0\\0\end{bmatrix}\right]$$

```
> Hmul(i,j);
```

$$\left[0, \begin{bmatrix}0\\0\\1\end{bmatrix}\right]$$

```
> Hscal(-1,e);
```

$$\left[-1, \begin{bmatrix}0\\0\\0\end{bmatrix}\right]$$

Question 2

```
> egal?:=(q1,q2)->evalb(q1[1]=q2[1]) and
  evalb(q1[2][1]=q2[2][1]) and evalb(q1[2][2]=q2[2][2]) and
  evalb(q1[2][3]=q2[2][3]):
> egal?(Hmul(e,[r,<x,y,z>]),[r,<x,y,z>]);
  egal?(Hmul([r,<x,y,z>],e),[r,<x,y,z>]);
```

true

true

```
> egal?(Hmul(i,i),Hscal(-1,e)); egal?(Hmul(j,j),Hscal(-1,e));
  egal?(Hmul(k,k),Hscal(-1,e));
```

true

true

true

```
> egal?(Hmul(i,j),k); egal?(Hmul(j,i),Hscal(-1,k));
  egal?(Hmul(j,k),i); egal?(Hmul(k,j),Hscal(-1,i));
  egal?(Hmul(k,i),j); egal?(Hmul(i,k),Hscal(-1,j));
```

true

true

true

true

true

true

Question 3

```
> res1:=Hmul([r1,<x1,y1,z1>],Hmul([r2,<x2,y2,z2>],[r3,<x3,y3,z3
  >]));
```

$$\begin{aligned}
res1 := [&r1\,(r2\,r3-x2\,x3-y2\,y3-z2\,z3)-x1\,(r3\,x2+r2\,x3+y2\,z3-z2\,y3)\\
&-y1\,(r3\,y2+r2\,y3+z2\,x3-x2\,z3)-z1\,(r3\,z2+r2\,z3+x2\,y3-y2\,x3),\\
&[(r2\,r3-x2\,x3-y2\,y3-z2\,z3)\,x1+r1\,(r3\,x2+r2\,x3+y2\,z3-z2\,y3)\\
&+y1\,(r3\,z2+r2\,z3+x2\,y3-y2\,x3)-z1\,(r3\,y2+r2\,y3+z2\,x3-x2\,z3)]\\
&[(r2\,r3-x2\,x3-y2\,y3-z2\,z3)\,y1+r1\,(r3\,y2+r2\,y3+z2\,x3-x2\,z3)\\
&+z1\,(r3\,x2+r2\,x3+y2\,z3-z2\,y3)-x1\,(r3\,z2+r2\,z3+x2\,y3-y2\,x3)]\\
&[(r2\,r3-x2\,x3-y2\,y3-z2\,z3)\,z1+r1\,(r3\,z2+r2\,z3+x2\,y3-y2\,x3)\\
&+x1\,(r3\,y2+r2\,y3+z2\,x3-x2\,z3)-y1\,(r3\,x2+r2\,x3+y2\,z3-z2\,y3)]]
\end{aligned}$$

```
> res2:=Hmul(Hmul([r1,<x1,y1,z1>],[r2,<x2,y2,z2>]),[r3,<x3,y3,z
  3>]);
```

$$\begin{aligned}
res2 := [&(r1\,r2-x1\,x2-y1\,y2-z1\,z2)\,r3-(r2\,x1+r1\,x2+y1\,z2-z1\,y2)\,x3\\
&-(r2\,y1+r1\,y2+z1\,x2-x1\,z2)\,y3-(r2\,z1+r1\,z2+x1\,y2-y1\,x2)\,z3,\\
&[r3\,(r2\,x1+r1\,x2+y1\,z2-z1\,y2)+(r1\,r2-x1\,x2-y1\,y2-z1\,z2)\,x3\\
&+(r2\,y1+r1\,y2+z1\,x2-x1\,z2)\,z3-(r2\,z1+r1\,z2+x1\,y2-y1\,x2)\,y3]\\
&[r3\,(r2\,y1+r1\,y2+z1\,x2-x1\,z2)+(r1\,r2-x1\,x2-y1\,y2-z1\,z2)\,y3\\
&+(r2\,z1+r1\,z2+x1\,y2-y1\,x2)\,x3-(r2\,x1+r1\,x2+y1\,z2-z1\,y2)\,z3]\\
&[r3\,(r2\,z1+r1\,z2+x1\,y2-y1\,x2)+(r1\,r2-x1\,x2-y1\,y2-z1\,z2)\,z3\\
&+(r2\,x1+r1\,x2+y1\,z2-z1\,y2)\,y3-(r2\,y1+r1\,y2+z1\,x2-x1\,z2)\,x3]]
\end{aligned}$$

```
> egal?(res1,res2);
```

false

```
> Map(normal,res1);
```

$$\begin{aligned}
[&-z1\,r3\,z2-z1\,r2\,z3-z1\,x2\,y3+z1\,y2\,x3-y1\,r3\,y2-y1\,r2\,y3-y1\,z2\,x3+y1\,x2\,z3\\
&-x1\,r3\,x2-x1\,r2\,x3-x1\,y2\,z3+x1\,z2\,y3+r1\,r2\,r3-r1\,x2\,x3-r1\,y2\,y3-r1\,z2\,z3,\\
&[-z1\,r3\,y2-z1\,r2\,y3-z1\,z2\,x3+z1\,x2\,z3+y1\,r3\,z2+y1\,r2\,z3+y1\,x2\,y3-y1\,y2\,x3\\
&+r1\,r3\,x2+r1\,r2\,x3+r1\,y2\,z3-r1\,z2\,y3+x1\,r2\,r3-x1\,x2\,x3-x1\,y2\,y3-x1\,z2\,z3]\\
&[-x1\,r3\,z2-x1\,r2\,z3-x1\,x2\,y3+x1\,y2\,x3+z1\,r3\,x2+z1\,r2\,x3+z1\,y2\,z3-z1\,z2\,y3\\
&+r1\,r3\,y2+r1\,r2\,y3+r1\,z2\,x3-r1\,x2\,z3+y1\,r2\,r3-y1\,x2\,x3-y1\,y2\,y3-y1\,z2\,z3]\\
&[-y1\,r3\,x2-y1\,r2\,x3-y1\,y2\,z3+y1\,z2\,y3+x1\,r3\,y2+x1\,r2\,y3+x1\,z2\,x3-x1\,x2\,z3\\
&+r1\,r3\,z2+r1\,r2\,z3+r1\,x2\,y3-r1\,y2\,x3+z1\,r2\,r3-z1\,x2\,x3-z1\,y2\,y3-z1\,z2\,z3]]
\end{aligned}$$

```
> Map(normal,res2);
```

$$[-z1\,r3\,z2-z1\,r2\,z3-z1\,x2\,y3+z1\,y2\,x3-y1\,r3\,y2-y1\,r2\,y3-y1\,z2\,x3+y1\,x2\,z3-x1\,r3\,x2-x1\,r2\,x3-x1\,y2\,z3+x1\,z2\,y3+r1\,r2\,r3-r1\,x2\,x3-r1\,y2\,y3-r1\,z2\,z3,$$
$$[-z1\,r3\,y2-z1\,r2\,y3-z1\,z2\,x3+z1\,x2\,z3+y1\,r3\,z2+y1\,r2\,z3+y1\,x2\,y3-y1\,y2\,x3+r1\,r3\,x2+r1\,r2\,x3+r1\,y2\,z3-r1\,z2\,y3+x1\,r2\,r3-x1\,x2\,x3-x1\,y2\,y3-x1\,z2\,z3]$$
$$[-x1\,r3\,z2-x1\,r2\,z3-x1\,x2\,y3+x1\,y2\,x3+z1\,r3\,x2+z1\,r2\,x3+z1\,y2\,z3-z1\,z2\,y3+r1\,r3\,y2+r1\,r2\,y3+r1\,z2\,x3-r1\,x2\,z3+y1\,r2\,r3-y1\,x2\,x3-y1\,y2\,y3-y1\,z2\,z3]$$
$$[-y1\,r3\,x2-y1\,r2\,x3-y1\,y2\,z3+y1\,z2\,y3+x1\,r3\,y2+x1\,r2\,y3+x1\,z2\,x3-x1\,x2\,z3+r1\,r3\,z2+r1\,r2\,z3+r1\,x2\,y3-r1\,y2\,x3+z1\,r2\,r3-z1\,x2\,x3-z1\,y2\,y3-z1\,z2\,z3]]$$

```
> egal?(Map(normal,res1),Map(normal,res2));
```

$$true$$

```
> Map(normal,Hadd(res1,Hscal(-1,res2)));
```

$$\left[0, \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}\right]$$

La loi est donc associative, puisque cette égalité dans H(Q(r_i,x_i,y_,z_i)) se transforme en une égalité dans H(R) en appliquant le morphisme évaluation

On pourrait de meme verifier la distributivité (à droite et à gauche) de la multiplication par rapport à l'addition, mais ces propriétés résultent directement des propriétés des produits scalaire et vectoriel.

```
> res1:=Hmul([r1,<x1,y1,z1>],Hadd([r2,<x2,y2,z2>],[r3,<x3,y3,z3
  >])):
> res2:=Hadd(Hmul([r1,<x1,y1,z1>],[r2,<x2,y2,z2>]),Hmul([r1,<x1
  ,y1,z1>],[r3,<x3,y3,z3>])):
> egal?(Map(normal,res1),Map(normal,res2));
```

$$true$$

Question 4

```
> N:=q->q[1]^2+VectorNorm(q[2],2,conjugate=false)^2:
> N([r,<a,b,c>]);
```

$$r^2+a^2+b^2+c^2$$

```
> Conj:=q->[q[1],VectorScalarMultiply(q[2],-1)]:
> Conj([r,<a,b,c>]);
```

$$\left[r, \begin{bmatrix} -a \\ -b \\ -c \end{bmatrix}\right]$$

```
> Inv:=q->Hscal(1/N(q),Conj(q)):
> Inv([r,<a,b,c>]);
```

$$\left[\frac{r}{r^2+a^2+b^2+c^2}, \begin{bmatrix} -\frac{a}{r^2+a^2+b^2+c^2} \\ -\frac{b}{r^2+a^2+b^2+c^2} \\ -\frac{c}{r^2+a^2+b^2+c^2} \end{bmatrix}\right]$$

```
> Map(normal,Hmul(%,[r,<a,b,c>]));
  Map(normal,Hmul([r,<a,b,c>],%%));
```

$$\left[1, \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}\right]$$

$$\left[1, \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}\right]$$

Question 5

```
> q0:=[r0,<x0,y0,z0>]: q:=[r,<x,y,z>]:
  Map(normal,Hadd(Hmul(q0,q),Hscal(-1,Hmul(q,q0))));
```

$$\left[0, \begin{bmatrix} 2\,y0\,z-2\,z0\,y \\ 2\,z0\,x-2\,x0\,z \\ 2\,x0\,y-2\,y0\,x \end{bmatrix}\right]$$

On reconnait les équations correspondant à la nullité du produit vectoriel de <x0,y0,z0> et <x,y,z> ; le centralisateur de q0 est donc l'ensemble des q non nuls de la forme [r,lambda<x0,y0,z0>].

Noter que l'équation précédente apparait directement lorsque l'on écrit l'égalité q0q=qq0 à la main : manipuler des expressions formelles par ordinateur est inutile.

Le centre est l'intersection des centralisateurs Zq, donc R^*

Question 6

```
> phi:=proc(q,h)
    if q=0 then error("q doit etre non nul") else
    return(Map(normal,Hmul(Hmul(q,h),Inv(q)))):fi:
  end:
> phi(0,i); phi(i,j);
Error, (in phi) q doit etre non nul
```

$$\left[0, \begin{bmatrix} 0 \\ -1 \\ 0 \end{bmatrix}\right]$$

```
> phi([r,<a,b,c>],[s,<x,y,z>]);
```

$$\left[s, \left[\begin{matrix} -\frac{-r^2 x-2 r b z+2 r c y-a^2 x-2 a b y-2 a c z+c^2 x+b^2 x}{r^2+a^2+b^2+c^2} \\ \frac{r^2 y+2 r c x-2 r a z+2 b a x+b^2 y+2 b c z-a^2 y-c^2 y}{r^2+a^2+b^2+c^2} \\ -\frac{-r^2 z-2 r a y+2 r b x-2 c a x-2 c b y-c^2 z+b^2 z+a^2 z}{r^2+a^2+b^2+c^2} \end{matrix}\right]\right]$$

```
> subs({x=0,y=0,z=0},%);
```

$$\left[s, \left[\begin{matrix} 0 \\ 0 \\ 0 \end{matrix}\right]\right]$$

Donc R est stable par phi_q et la restriction de phi_q à R est l'identité

```
> subs(s=0,%%);
```

$$\left[0, \left[\begin{matrix} -\frac{-r^2 x-2 r b z+2 r c y-a^2 x-2 a b y-2 a c z+c^2 x+b^2 x}{r^2+a^2+b^2+c^2} \\ \frac{r^2 y+2 r c x-2 r a z+2 b a x+b^2 y+2 b c z-a^2 y-c^2 y}{r^2+a^2+b^2+c^2} \\ -\frac{-r^2 z-2 r a y+2 r b x-2 c a x-2 c b y-c^2 z+b^2 z+a^2 z}{r^2+a^2+b^2+c^2} \end{matrix}\right]\right]$$

Donc E est stable par phi_q

Question 7

```
> q:=[r,<a,b,c>]:
> C1:=phi(q,i)[2]: C2:=phi(q,j)[2]: C3:=phi(q,k)[2]:
> M:=<C1|C2|C3>;
```

$$M := \left[\begin{matrix} -\frac{-r^2-a^2+c^2+b^2}{r^2+a^2+b^2+c^2} & \frac{2(-r c+a b)}{r^2+a^2+b^2+c^2} & \frac{2(r b+a c)}{r^2+a^2+b^2+c^2} \\ \frac{2(r c+a b)}{r^2+a^2+b^2+c^2} & \frac{r^2+b^2-a^2-c^2}{r^2+a^2+b^2+c^2} & \frac{2(-r a+b c)}{r^2+a^2+b^2+c^2} \\ -\frac{2(r b-a c)}{r^2+a^2+b^2+c^2} & \frac{2(r a+b c)}{r^2+a^2+b^2+c^2} & -\frac{-r^2-c^2+b^2+a^2}{r^2+a^2+b^2+c^2} \end{matrix}\right]$$

Vérifions que rho(q) appartient a SO3(R) :

```
> Map(normal,Transpose(M).M); Map(normal,M.Transpose(M));
  Determinant(M);
```

$$\left[\begin{matrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{matrix}\right]$$

$$\left[\begin{matrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{matrix}\right]$$

$$1$$

Noter que le calcul des points fixes suffit, sachant qu'on a affaire à une isométrie : si q=[r,u], c'est la droite Ru, puisque qhq^(-1)=q est équivalent au fait que q et h commutent (on a déjà calculé le centralisateur). Il s'agit donc d'une rotation d'axe Ru.

```
> phi(q,[r,<a,b,c>]);
```

$$\left[r, \left[\begin{matrix} a \\ b \\ c \end{matrix}\right]\right]$$

Question 8

```
> v1:=<a,b,c>: v1:=v1/VectorNorm(v1,2,conjugate=false):
  v2:=<b,-a,0>: v2:=v2/VectorNorm(v2,2,conjugate=false):
  v3:=CrossProduct(v1,v2):
  v3:=v3/VectorNorm(v3,2,conjugate=false): P:=<v1|v2|v3>;
```

$$P := \left[\begin{matrix} \frac{a}{\sqrt{a^2+b^2+c^2}}, & \frac{b}{\sqrt{b^2+a^2}}, & c\,a \Big/ \left(\left(\frac{c^2 a^2}{(a^2+b^2+c^2)(b^2+a^2)}+\frac{c^2 b^2}{(a^2+b^2+c^2)(b^2+a^2)}+\left(-\frac{a^2}{\sqrt{a^2+b^2+c^2}\sqrt{b^2+a^2}}-\frac{b^2}{\sqrt{a^2+b^2+c^2}\sqrt{b^2+a^2}}\right)^2\right)^{(1/2)} \sqrt{a^2+b^2+c^2}\,\sqrt{b^2+a^2}\right) \\ \frac{b}{\sqrt{a^2+b^2+c^2}}, & -\frac{a}{\sqrt{b^2+a^2}}, & c\,b \Big/ \left(\left(\frac{c^2 a^2}{(a^2+b^2+c^2)(b^2+a^2)}+\frac{c^2 b^2}{(a^2+b^2+c^2)(b^2+a^2)}+\left(-\frac{a^2}{\sqrt{a^2+b^2+c^2}\sqrt{b^2+a^2}}-\frac{b^2}{\sqrt{a^2+b^2+c^2}\sqrt{b^2+a^2}}\right)^2\right)^{(1/2)} \sqrt{a^2+b^2+c^2}\,\sqrt{b^2+a^2}\right) \\ \frac{c}{\sqrt{a^2+b^2+c^2}}, & 0, & \left(-\frac{a^2}{\sqrt{a^2+b^2+c^2}\sqrt{b^2+a^2}}-\frac{b^2}{\sqrt{a^2+b^2+c^2}\sqrt{b^2+a^2}}\right) \Big/ \left(\frac{c^2 a^2}{(a^2+b^2+c^2)(b^2+a^2)}+\frac{c^2 b^2}{(a^2+b^2+c^2)(b^2+a^2)}+\left(-\frac{a^2}{\sqrt{a^2+b^2+c^2}\sqrt{b^2+a^2}}-\frac{b^2}{\sqrt{a^2+b^2+c^2}\sqrt{b^2+a^2}}\right)^2\right)^{(1/2)} \end{matrix}\right]$$

C'est la matrice de changement de base. Les coefficients sont dans Q(a,b,c)(d,e) où d^2=a^2+b^2 et e^2=a^2+b^2+c^2. Il s'agit d'une extension de corps de degré 4 du corps Q(a,b,c) des fractions rationnelles en 3 indéterminées.

Simplifions :

```
> P:=Map(normal,P);
```

$$P := \begin{bmatrix} \frac{a}{\sqrt{a^2+b^2+c^2}} & \frac{b}{\sqrt{b^2+a^2}} & \frac{c\,a}{\sqrt{a^2+b^2+c^2}\,\sqrt{b^2+a^2}} \\ \frac{b}{\sqrt{a^2+b^2+c^2}} & -\frac{a}{\sqrt{b^2+a^2}} & \frac{c\,b}{\sqrt{a^2+b^2+c^2}\,\sqrt{b^2+a^2}} \\ \frac{c}{\sqrt{a^2+b^2+c^2}} & 0 & -\frac{\sqrt{b^2+a^2}}{\sqrt{a^2+b^2+c^2}} \end{bmatrix}$$

```
> M1:=Map(normal,Transpose(P).M.P);
```

$$M1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & -\frac{-r^2+a^2+b^2+c^2}{r^2+a^2+b^2+c^2} & -\frac{2\sqrt{a^2+b^2+c^2}\,r}{r^2+a^2+b^2+c^2} \\ 0 & \frac{2\sqrt{a^2+b^2+c^2}\,r}{r^2+a^2+b^2+c^2} & -\frac{-r^2+a^2+b^2+c^2}{r^2+a^2+b^2+c^2} \end{bmatrix}$$

Le changement de base a un sens dans M(R) lorsque (a,b)<>(0,0), étant donné le choix du vecteur v2. Dans le cas contraire, on peut prendre v2=<0,c,-b>. Cela revient à effectuer une permutation circulaire des 3 composantes qui jouent un role symétrique, donc il est inutile de considérer un second cas. Bien entendu, le cas (a,b,c)=(0,0,0) correspond à l'identité.

Question 9

```
> hyp:=r^2+a^2+b^2+c^2=1:
> subs(hyp,M1); algsubs(hyp,M1);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & r^2-a^2-b^2-c^2 & -2\sqrt{a^2+b^2+c^2}\,r \\ 0 & 2\sqrt{a^2+b^2+c^2}\,r & r^2-a^2-b^2-c^2 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 2\,r^2-1 & -2\sqrt{-r^2+1}\,r \\ 0 & 2\sqrt{-r^2+1}\,r & 2\,r^2-1 \end{bmatrix}$$

Attention, Maple ne privilégie pas forcément la variable a pour l'élimination ; on peut obtenir (essayer d'exécuter cette feuille de calculs plusieurs fois) :

```
> algsubs(hyp,M1,[r]);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & -2\,a^2-2\,b^2-2\,c^2+1 & -2\sqrt{a^2+b^2+c^2}\,r \\ 0 & 2\sqrt{a^2+b^2+c^2}\,r & -2\,a^2-2\,b^2-2\,c^2+1 \end{bmatrix}$$

Donc on utilise plutot :

```
> M1:=algsubs(hyp,M1,[a]);
```

$$M1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 2\,r^2-1 & -2\sqrt{-r^2+1}\,r \\ 0 & 2\sqrt{-r^2+1}\,r & 2\,r^2-1 \end{bmatrix}$$

```
> subs(r=cos(t),M1);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 2\cos(t)^2-1 & -2\sqrt{-\cos(t)^2+1}\,\cos(t) \\ 0 & 2\sqrt{-\cos(t)^2+1}\,\cos(t) & 2\cos(t)^2-1 \end{bmatrix}$$

```
> M1:=algsubs(r=cos(t),M1);
```

$$M1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 2\cos(t)^2-1 & -2\sqrt{-\cos(t)^2+1}\,\cos(t) \\ 0 & 2\sqrt{-\cos(t)^2+1}\,\cos(t) & 2\cos(t)^2-1 \end{bmatrix}$$

```
> M1:=Map(expr->simplify(expr,trig),M1);
```

$$M1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 2\cos(t)^2-1 & -2\sqrt{\sin(t)^2}\,\cos(t) \\ 0 & 2\sqrt{\sin(t)^2}\,\cos(t) & 2\cos(t)^2-1 \end{bmatrix}$$

```
> assume(t>=0,t<=Pi);
> M1:=Map(expr->simplify(expr,trig),M1);
```

$$M1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 2\cos(t\sim)^2-1 & -2\sin(t\sim)\cos(t\sim) \\ 0 & 2\sin(t\sim)\cos(t\sim) & 2\cos(t\sim)^2-1 \end{bmatrix}$$

```
> M1:=Map(expr->combine(expr,trig),M1);
```

$$M1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos(2\,t\sim) & -\sin(2\,t\sim) \\ 0 & \sin(2\,t\sim) & \cos(2\,t\sim) \end{bmatrix}$$

En s'y prenant bien, on trouve 2t=theta_q. Sur cet exemple, il est plus simple de manipuler les expressions trigonométriques à la main.

Question 10

Si q1=j on trouve la rotation r1 d'axe R<0,1,0> et d'angle pi (on résoud cos t=0 et sin t=1 ; l'angle vaut 2t)

Réciproquement, la rotation r2 d'angle pi/2 et d'axe R(i+j) correspond (modulo R) à un quaternion de la forme q2=[r,lambda<1,1,0>] avec r=cos pi/4=sqrt(2)/2 et lambda sqrt(2)=sqrt(2)/2. On peut prendre q2=[sqrt(2),<1,1,0>].

Finalement, la composée r1.r2 correspond au produit q1.q2 :

```
> q3:=Hmul(j,[sqrt(2),<1,1,0>]); Hscal(1/sqrt(N(q3)),q3);
```

$$q3 := \left[-1, \begin{bmatrix} 0 \\ \sqrt{2} \\ -1 \end{bmatrix} \right]$$

$$\left[\frac{-1}{2}, \begin{bmatrix} 0 \\ \frac{\sqrt{2}}{2} \\ \frac{-1}{2} \end{bmatrix} \right]$$

C’est la rotation d’axe R<0,sqrt(2),-1> et d’angle 2pi/3 (on résoud cost t=-1/2 et sint t=sqrt(3)/2).